

SAP system から Brother P-touch/QL label printer への印字方法

1 概要

この資料は、ブラザーP-touch/QL ラベルプリンタを SAP R/3 で使用するために必要なソフトのインストール方法と使用方法について説明したものです。

(対象となるプリンタの例. QL-550、QL-650TD、PT-9500pc)

ブラザーP-touch/QL ラベルプリンタを SAP R/3 で使用するには、ブラザーの b-PAC SDK(ソフトウェア開発キット)、P-touch Editor で作成したテンプレートファイル、対象プリンタドライバが必要です。

資料で説明する SAP R/3 プログラムは、プリンタを起動するプログラムとなります。

2 動作概要

SAP R/3 上からブラザーP-touch/QL ラベルプリンタへラベル印刷するステップは、以下のようになります。

- SAP R/3 のプログラムより、OLE 機能を実行します。
- OLE 機能の実行により、b-PAC コンポーネントが使用されます。
- SAP R/3 よりデータを抽出し、b-PAC 経由でテンプレートの文字などを入れ替え、P-touch ラベルプリンタにデータが転送されます。
- P-touch/QL プリンタで印刷が実行されます。

3 ご利用にあたって

ブラザーP-touch/QL プリンタを使用するには、最初に以下の手順を実行する必要があります。

- 最新のブラザーP-touch/QL ラベルプリンタをセットアップします。
 - プリンタドライバおよび P-touch Editor をインストールします。(ラベルプリンタに同梱されている CD を使用するか、Brother Solutions Center からダウンロードします。)
 - P-touch Editor を使用して、使用するプリンタ、ラベルの種類を選択し、テキストやバーコードなどを配置して、b-PAC から呼び出すテンプレートファイルを作成します。
 - P-touch Editor はテンプレートの作成のみにします。出力するプリンタやラベルデザインを変更される場合は、P-touch Editor で再度テンプレートファイルを編集します。テンプレートの作成方法の詳細は、b-PAC SDK を参照願います。
 - SAP R/3 から印刷テストを行う前に、P-touch Editor から作成したテンプレートが印刷されるかを確認してください。
- 最新のブラザーb-PACSDK をインストールします。

これは OLE 機能を利用して、SAP R/3 からブラザーP-touch/QL ラベルプリンタへ出力するためのソフトウェア開発キット(SDK)で、SAP R/3 から出力するための COM コンポーネント、ドキュメントおよびサンプルを含んでいます。開発用の PC ではなく、各クライアント PC にインストールされる場合は、b-PAC クライアントインストーラーを用意しておりますので、こちらをご利用願います。

4 作業手順

1) C: \ptlabel フォルダを出力を行いたいローカル PC に作成してください。

このフォルダに EXAMPLE.lbl をコピーします。

C: \ptlabel はデフォルトフォルダとして使用します。

2) SAP R/3 にて OLE 機能を使用する為の定義を設定して下さい。(設定方法は後述(5.1)参照)

3) SAP R/3 にてレポートプログラムを作成して下さい。

テキストファイル ZA_CA_PTLABEL_EXAMPLE.txt をアップロードするか、もしくは、後述のサンプルプログラムの内容をコピー&ペーストして下さい。

プログラム名 ZA_CA_PTLABEL_EXAMPLE

4) SAP R/3 レポートプログラム ZA_CA_PTLABEL_EXAMPLE を起動させます。


補足

- テンプレート上の様々なオプションを変更することができます。例えばテキストの位置の変更、文字スタイルの変更、文字サイズの変更、テープサイズの変更、使用するプリンタの変更など。これらの変更は P-touch Editor で"EXAMPLE.lbl"ファイルを開き、テンプレートを編集することで実現できます。
- テキストオプションを枠サイズ固定にすることで、テキストデータの長さに関わらず、常に同じサイズのラベルにすることが出来ます。(テキストの幅はテキスト枠の大きさにあうよう自動的に調整されます。) またテープ長を自動し、テキストのオプションを自由長にすることで、テキストの長さに合わせて、ラベルの長さを変更することもできます。

5 SAP R/3 上での OLE 機能の定義方法

R/3 から OLE 機能を使用する場合は、R/3 内のテーブルにオブジェクト名とクラス ID を定義しなければなりません。以下に、その手順を説明します。

- ・アプリケーション名の登録
トランザクションコード”SOLE”を実行します。
ここで、”Display -> Change (照会-> 変更)”ボタンを選択します。




Display View "Maintenance view for OLE applications": Overview

Table View Edit Goto Selection Utilities(M) System Help

OLE application	CLSID	Check auth.
ALVIEWER.APP	{30446849-11E3-11D0-9D4E-0020AFC0E4CC}	<input type="checkbox"/>
APPROACH.APPLICATION	{00028701-0000-0000-C000-000000000046}	<input type="checkbox"/>
AUTOVUE.APPLICATION	{63A20B20-2357-101D-909B-0000C02E4A4A}	<input type="checkbox"/>
AUTOVUE.DOCUMENT	{71FE6890-6F6E-11CE-9EF9-0000C074797F}	<input type="checkbox"/>

新規の場合は、”New entries (新規エントリ)” ボタンを選択します。




Change View "Maintenance view for OLE applications": Overview

Table View Edit Goto Selection Utilities(M) System Help

OLE application	CLSID	Check auth.
ALVIEWER.APP	{30446849-11E3-11D0-9D4E-0020AFC0E4CC}	<input type="checkbox"/>
APPROACH.APPLICATION	{00028701-0000-0000-C000-000000000046}	<input type="checkbox"/>
AUTOVUE.APPLICATION	{63A20B20-2357-101D-909B-0000C02E4A4A}	<input type="checkbox"/>
AUTOVUE.DOCUMENT	{71FE6890-6F6E-11CE-9EF9-0000C074797F}	<input type="checkbox"/>

以下の項目を登録します。

OLE application : BRSSCOM.DOCUMENT
Version number : 1.44
CLSID : {7979FAA5-3A2A-4E50-A2C9-286C6F34C1D3}
CLSID LibType : {7979FAA5-3A2A-4E50-A2C9-286C6F34C1D3}
OLE object name : DOCUMENT
Type Info key : NO_TYPELIB
Text : Brother P-touch Applicable Component



上に示した画面のように項目の入力が終わったら、'Save(保存)'ボタンを押して登録します。


5.1 SAP 起動画面


SAP R/3 からのプログラム起動例

プログラム `ZA_CA_PTLABEL_EXAMPLE` 起動

デフォルト値の設定がしてあります。値は変更可能です。

ただし、*Template File path* には、ラベルレイアウトの保存先を指定して下さい。




実行ボタン  (F8)を押下して下さい。



The screenshot shows a SAP program execution window titled "check the print out of P-touch and QL labels from SAP". The window has a menu bar with "Program", "Edit", "Goto", "System", and "Help". Below the menu bar is a toolbar with various icons. The main area of the window is a data entry form with the following fields and values:

Template File path	C:\PTLabel\EXAMPLE.lbl
Sales Organization	BIE
Customer no.	11111
Customer Name	Brother Test example
Material	DCP110
Plant	B01
Storage Location	B110
Unrestricted Quantity	25
Available quantity	22
Rate	50.00
Rate	52.00
Storage Bin	L100
Material Description	DCP110C MULTI FUNCTION
EAN/UPC	4977766627085

下記がラベルプリント実行結果です。
バーコード変更は P-touch ラベルレイアウトにて操作できます。

Customer No. 1111		BROTHER TEST EXAMPLE	
		Cust.Price	
11111		52.00	
Material DCP110		DCP110C MULTI FUNCTION	
Plant	B01		Stock 25
S.Loc	B110		Avail.Qty 22
StorBin	L100	*DCP110*	Z001Price 50.00
EAN number			
			User MARKUSU
4977766627085			Date 02.09.2005
			Time 12:16:17

6 SAP R/3 プログラムソースコードサンプル

```
*---DESCRIPTION-----
* check the print out of P-touch and QL labels from SAP
*---CHANGE HISTORY-----
* NO DATE    PROGRAMMER    DESCRIPTION
*
*-----
REPORT ZA_CA_PTLABEL_EXAMPLE01 NO STANDARD PAGE HEADING.

*-- include
  INCLUDE ole2incl.

*-- Work areas
DATA: OBJDOC    TYPE OLE2_OBJECT,
      WA_SET    TYPE I,
      WA_COUNT  TYPE I,
      WA_BAR    TYPE I,
      N         TYPE I,
      WA_INDEX  TYPE I,
      STR(56),
      WA_INPUT(56),
      WA_TEXT(8),
      F_ERR(1).

*-- Selection screen
* An initial screen
* set default data for indication item
Parameters:
  p_str(56) default 'C:\PTLabel\EXAMPLE.lbl'.
Parameters:
  P_VKO type VKORG default 'BIE',
  P_kun(12) default '11111',
  P_name(42) default 'Brother Test example',
  P_MATNR LIKE mara-matnr default 'DCP110', "A Material
  P_WERKS LIKE t001w-werks default 'B01',  "B Plant
  P_LGORT LIKE t001l-igort default 'B110',  "C St. Location
  P_labst(17) default '25',                "D Stock
  P_avqty(17) default '22',                "E Avlble Qty
  P_zkbetr LIKE konp-kbetr default '50.00', "F Prices Z001
  P_ckbetr LIKE konp-kbetr default '52.00', "G Prices custome
  P_lgpla LIKE lqua-lgpla default 'L100',
          " Location/storage bin "1A
  P_maktx LIKE makt-maktx default 'DCP110C MULTI FUNCTION',
          "I Material Description
  P_EAN11 LIKE mara-EAN11 default '4977766627085'. "J EAN code

*-- main process
start-of-selection.

  CLEAR: F_ERR.
*-- create object for using label print program
  CREATE OBJECT OBJDOC 'BrssCom.Document'.
  IF SY-SUBRC <> 0.
    WRITE 'Object:BrssCom.Document Create ERROR'.
  FREE OBJECT OBJDOC.
  EXIT.
ENDIF.
```

```

*-- template file of label is open
CALL METHOD OF OBJDOC 'OPEN' = WA_SET EXPORTING #1 = p_str.
IF WA_SET = 0.
  WRITE 'File Open ERROR'.
  FREE OBJECT OBJDOC.
  EXIT.
ENDIF.

```

```

*-- read label's template file
CALL METHOD OF OBJDOC 'Gettextcount' = WA_COUNT.
IF WA_COUNT < 0.
  WRITE 'Template file Read ERROR'.
  FREE OBJECT OBJDOC.
  EXIT.
ENDIF.

```

```

*-- data set for label's template file
*-----*
** this 'do' process is column of output label that meaning is
** nothing, just follow label's template file
*-----*
* 17 is count of setting data for label
DO 17 TIMES.
  CLEAR: WA_INPUT, WA_TEXT.
  N = SY-INDEX. "Count of do loop

```

```

*-- get column name of template file and output data for label
* WA_INPUT is template file column name
* WA_TEXT is output data for label
  PERFORM DATASET USING N
  CHANGING WA_INPUT
  WA_TEXT.

```

```

*-- get template file column index
CALL METHOD OF OBJDOC 'GetTextIndex' = WA_INDEX
EXPORTING #1 = WA_TEXT : .

```

```

*-- When counts are 14 times or less
* Less than 14 is normal data
* More than 14 is barcode data
  IF N <= 14. "When counts are 14 times or less

```

```

*-- set output data(normal data) for label
CALL METHOD OF OBJDOC 'SetText' = WA_SET
EXPORTING #1 = WA_INDEX :
          #2 = WA_INPUT.

```

```

*-- error handling
* 'SetText'method is failed, show error message
IF WA_SET = 0. "error
  WRITE 'Set Text ERROR'.
  F_ERR = 'X'. "error flag on
  EXIT.
ENDIF.

```

ELSE.

```

*-- when the count is larger than 14 times
*-- set output data(barcode data) for label
CALL METHOD OF OBJDOC 'SetBarcodedata' = WA_BAR
EXPORTING #1 = WA_TEXT :
          #2 = WA_INPUT.

```

```
IF WA_BAR = 0.          "error
  WRITE 'Set Barcodedata ERROR'.
  F_ERR = 'X'.          "error flag on
  EXIT.
ENDIF.
```

```
ENDIF.
ENDDO.
```

```
*-- Print for label
IF F_ERR = SPACE.  "case error flag on
  CALL METHOD OF OBJDOC 'DoPrint'
    EXPORTING #1 = 0 :
              #2 = '0'.
ENDIF.
```

```
*-- object of label print program is Release
FREE OBJECT OBJDOC.
```

```
*-----*
* Form DATASET
*---DESCRIPTION-----*
* create template file's column name and output data for label.
* this 'P_INDEX(1 to 14)' is column count of initial screen
* that meaning is nothing,
* just turn from top of initial screen column.
* this 'P_INDEX(15 to 17)' meaning is nothing just for program.
* ↓ Complement
* 'Text0' to 'Text17' setting column name of label's template file.
* if you want to change column name of label's template file,
* you should change setting value of P_TEXT.
*-----*
* Using parameter P_INDEX index of parameter count
* Changing parameter P_INPUT output data for label
* P_TEXT column name of label's template file
*-----*
FORM dataset USING VALUE(P_INDEX)
  CHANGING VALUE(P_INPUT)
  VALUE(P_TEXT).
```

```
*-- set data
CASE P_INDEX.
```

```
* 1 to 14 is output data(normal data)
  WHEN 1.
    P_INPUT = P_kun.          "Customernumber
    P_TEXT = 'Text0'.

  WHEN 2.
    P_INPUT = P_name.        "Customer name

    P_TEXT = 'Text1'.

  WHEN 3.
    WRITE SY-DATUM TO P_INPUT.  "Date
    P_TEXT = 'Text2'.

  WHEN 4.
    WRITE SY-TIMLO TO P_INPUT.  "Time
```

```

P_TEXT = 'Text3'.

WHEN 5.
P_INPUT = SY-UNAME.           "Username
P_TEXT = 'Text4'.

WHEN 6.
P_INPUT = P_MATNR.           "Materialnumber
P_TEXT = 'Text5'.

WHEN 7.
P_INPUT = P_WERKS.           "Plant
P_TEXT = 'Text6'.
WHEN 8.
P_INPUT = P_LGORT.           "Storage location
P_TEXT = 'Text7'.

WHEN 9.
P_INPUT = P_labst.           "Stock
P_TEXT = 'Text8'.

WHEN 10.
P_INPUT = P_avqty.           "AvailQuantity
P_TEXT = 'Text9'.

WHEN 11.
WRITE P_zkbetr TO P_INPUT LEFT-JUSTIFIED. "Z001Price
P_TEXT = 'Text10'.

WHEN 12.
WRITE P_ckbetr TO P_INPUT LEFT-JUSTIFIED. "CustomerPrice
P_TEXT = 'Text11'.

WHEN 13.
P_INPUT = P_lgpla.           "StorageBin
P_TEXT = 'Text12'.

WHEN 14.
P_INPUT = P_maktx.           "Material name
P_TEXT = 'Text13'.
* 15 to 17 is output data (barcode data)
WHEN 15.
P_INPUT = P_EAN11.           "EAN number
P_TEXT = '0'.

WHEN 16.
P_INPUT = P_MATNR.           "Materialnumber
P_TEXT = '1'.

WHEN 17.
P_INPUT = P_kun.             "Customernumber
P_TEXT = '2'.

ENDCASE.

ENDFORM.                   " DATASET

```